变速箱决策管理模块整机信息程序设计

一、先创建“整机库存计算”窗体“frmzjkc_cal_child”：

按照下面的属性设置控件：

写代码：

Private Sub cmdOK_Click()

'核算整机库存

    Dim Month1 As String    '月份

    Dim date1 As String    '月初第一天

    Dim date2 As String    '月末最后一天

    Dim strsql1 As String

    Dim strsql2 As String

    Dim strsql3 As String

    Dim strsql4 As String

    Dim strsql5 As String

    date1 = "#" & Me.sDate & "#"    '月初第一天

    date2 = "#" & Me.eDate & "#"    '月末最后一天

    '清空 整机库存计算的临时表 tblZjxx_temp表

    DoCmd.RunSQL "DELETE tblZjxx_temp.* FROM tblZjxx_temp;"

    '求出上个月的月份

    Month1 = "#" & DMax("[kcyf]", "tblZjkcyj") & "#"

    If Month1 = "##" Then    '当没有一条记录时

        MsgBox "第一次使用请输入整机的上期结存明细！", vbCritical, "提示"

        Exit Sub

    End If

    DoCmd.SetWarnings False

    '将月初整机库存明细追加到表 tblZjxx_temp

    strsql1 = "INSERT INTO tblZjxx_temp ( jxid, ycjc ) " _

            & "SELECT tblZjkcyj.jxID, tblZjkcyj.kcl " _

            & "FROM tblZjkcyj " _

            & "WHERE tblZjkcyj.kcyf= " & Month1

    DoCmd.RunSQL strsql1

    '将本月订货整机明细追加到表 tblZjxx_temp

    strsql2 = "INSERT INTO tblZjxx_temp ( jxid, ddts ) " _

            & "SELECT tblXsddzj.jxID, tblXsddzj.zjsl " _

            & "FROM tblXsddzj " _

            & "WHERE tblXsddzj.dhrq Between " & date1 & " And " & date2

    DoCmd.RunSQL strsql2

    '将本月装配整机明细追加到表 tblZjxx_temp

    strsql3 = "INSERT INTO tblZjxx_temp ( jxid, zpts ) " _

            & "SELECT tblZjjcmx.jxID, tblZjjcmx.jcsl " _

            & "FROM tblZjjcmx " _

            & "WHERE tblZjjcmx.jcrq Between " & date1 & " And " & date2

    DoCmd.RunSQL strsql3

    '将本月出仓整机明细追加到表 tblZjxx_temp

    strsql4 = "INSERT INTO tblZjxx_temp ( jxid, ckts ) " _

            & "SELECT tblZjccmx.jxID, tblZjccmx.ccsl " _

            & "FROM tblZjccmx " _

            & "WHERE tblZjccmx.ccrq Between " & date1 & " And " & date2

    DoCmd.RunSQL strsql4

    '将tblZjxx_temp表按机型代码进行求和，并将结果写入到整机信息表tblZjxx中

    '清空 整机库存计算的临时表 tblZjxx_temp表

    DoCmd.RunSQL "DELETE tblZjxx.* FROM tblZjxx;"    '先清空tblZjxx表

    strsql5 = "INSERT INTO tblZjxx ( jxid, ycjc, ddts, zpts, ckts, brkc ) " _

            & "SELECT tblZjxx_temp.jxid, Sum(tblZjxx_temp.ycjc) AS yc, Sum(tblZjxx_temp.ddts) AS dd, Sum(tblZjxx_temp.zpts) AS zp, Sum(tblZjxx_temp.ckts) AS ck, [yc]+[zp]-[ck] AS br " _

            & "FROM tblZjxx_temp " _

            & "GROUP BY tblZjxx_temp.jxid;"

    DoCmd.RunSQL strsql5

    Me.Refresh

    '加载整机信息子窗体

    Forms!usysfrmMain!frmChild.SourceObject = "frmZjxx_child"

    '让子窗体凹陷

    Forms!usysfrmMain!frmChild.SpecialEffect = 2

End Sub
Private Sub Form_Load()

'窗体一加载，默认为月初第一天和最后一天

    Me.sDate.SetFocus

    Me.sDate = DateSerial(Year(Date), Month(Date), 1)

    Me.eDate.SetFocus

    Me.eDate = DateSerial(Year(Date), Month(Date) + 1, 1) - 1

    Me.cmdOK.SetFocus

End Sub

Private Sub sDate_AfterUpdate()

    Me.eDate = DateSerial(Year(Me.sDate), Month(Me.sDate) + 1, 1) - 1

End Sub
二、整机信息查询：qryzjxx的创建：

然后以此查询为数据源，创建整机信息子窗体：frmzjxx_child:

窗体属性设置：

窗体代码：

Private Sub Form_Close()

    Forms!usysfrmMain.cmdJzzj.Visible = False

End Sub

Private Sub Form_Open(Cancel As Integer)

'点击查找按钮时，打开自定义的查找窗体

    Forms!usysfrmMain!labFind.Tag = 1

    '让整机库存结转按钮显示

    Forms!usysfrmMain.cmdJzzj.Visible = True

    '调整按钮的位置，放在最右边

    Forms!usysfrmMain.cmdJzzj.Left = Forms!usysfrmMain.InsideWidth - Forms!usysfrmMain.cmdJzzj.Width - 100

End Sub

Public Sub FindEnd()

    Forms!usysfrmMain!frmChild.Form.RecordSource = Acchelp_ChildFormRecordSource("qryZjxx", "机型代码", True)

End Sub

Public Sub btnFind()

'打开查询窗体

    DoCmd.OpenForm "usysfrmFind"

    '给查询窗体的字段选择赋值 字段名称后跟字段类型：1-日期或时间；2-数字；3-文本

    '下方的字段名称来自查询 qryZjxx

    Forms!usysfrmFind!cobfldName.RowSourceType = "值列表"

    Forms!usysfrmFind!cobfldName.RowSource = "机型代码;3;整机名称;3;上月结存;2;本月订货台数;2;本月装配台数;2;本月出库台数;2;本日库存;2;"

    '指定数据来源为 qryZjxx

    Forms!usysfrmFind!labDataSource.Caption = "qryZjxx"

End Sub

Public Sub btnPrint()

'以预览方式打开报表rptZjxx

    DoCmd.OpenReport "rptZjxx", acViewPreview

End Sub

Public Sub btnSave()

DoCmd.OpenForm "frmZjMonth"

End Sub
三、结转月份窗体frmzjmonth的创建：

代码编写：

Private Sub cmdClose_Click()

DoCmd.Close

End Sub

Private Sub cmdOK_Click()

    Dim strMonth As String

    strMonth = "#" & Me.txtMonth & "#"

    If MsgBox("您确认要结转本月整机库存吗?", vbYesNo + vbInformation, Forms!usysfrmLogin.Caption) = vbYes Then

        DoCmd.SetWarnings False

        DoCmd.RunSQL "INSERT INTO tblZjkcyj ( jxID, kcl, kcyf ) " _

                   & "SELECT tblZjxx.jxid, tblZjxx.brkc, " & strMonth & " AS kcMonth " _

                   & "FROM tblZjxx"

                   MsgBox "整机库存结转成功！", vbInformation, Forms!usysfrmLogin.Caption

                   DoCmd.Close acForm, "frmZjMonth"

    End If

End Sub
四、论坛中库存结转的详细制作教程（来自论坛）：

变速箱库存结转
在专业版的变速箱教程的学习中，当我们做到决策管理部分，我们需要添加一个库存结转的功能，这个库存结转功能我们可以通过一个追加查询来实现，下面来具体讲一下实现过程。

首先，先把教程中的决策管理中的整机信息部分的功能做好。然后我们在平台的usysfrmmain窗体上增加一个按钮，进入usysfrmmain窗体的设计视图，在适当的位置添加一个按钮控件，效果如下图所示：

红色框内即为我们添加的按钮，我们把它命名为cmdJzzj，并把这个按钮的“可见性”属性设置为“否”，“标题”属性设置为“整机库存结转”。然后我们进入frmZjxx_child窗体，在窗体的加载事件中写代码如下：

Private Sub Form_Open(Cancel As Integer)

'点击查找按钮时，打开自定义的查找窗体

    Forms!usysfrmMain!labFind.Tag = 1

    '让整机库存结转按钮显示

    Forms!usysfrmMain.cmdJzzj.Visible = True

    '调整按钮的位置，放在最右边

    Forms!usysfrmMain.cmdJzzj.Left = Forms!usysfrmMain.InsideWidth - Forms!usysfrmMain.cmdJzzj.Width - 100

End Sub
注意红色代码部分，第一行红色代码是把我们刚才在usysfrmmain窗体上添加的按钮的“可见性”属性设置为“是”，下面的代码是设置按钮的位置在最右边。当我们设置完查询条件按了确定按钮后，打开frmZjxx_child窗体的时候，我们添加的按钮就会在usysfrmmain窗体上显示出来了。

现在我们来添加一个结转的窗体，新建一个空白窗体，在窗体里添加一个文本框命名为txtMonth，一个列表框命名为lstMont，2个按钮，分别命名为cmdOK和cmdClose，按钮标题属性设置为“确定(&O)”和“关闭(&C)”，然后保存窗体，命名为frmZjMonth，整个界面完成后如下图：

按照新增，修改窗体把该窗体的属性设置一下，然后列表框的行来源属性设置为SELECT tblZjkcyj.kcyf FROM tblZjkcyj GROUP BY tblZjkcyj.kcyf ORDER BY tblZjkcyj.kcyf DESC WITH OWNERACCESS OPTION;，这样是为了在列表中显示tblZjkcyj中所有已经结存过的月份。在确定按钮的单击事件中写代码如下：

Private Sub cmdOK_Click()

    Dim strMonth As String

    strMonth = "#" & Me.txtMonth & "#"

    If MsgBox("您确认要结转本月整机库存吗?", vbYesNo + vbInformation, Forms!usysfrmLogin.Caption) = vbYes Then

        DoCmd.SetWarnings False

        DoCmd.RunSQL "INSERT INTO tblZjkcyj ( jxID, kcl, kcyf ) " _

                   & "SELECT tblZjxx.jxid, tblZjxx.brkc, " & strMonth & " AS kcMonth " _

                   & "FROM tblZjxx"

                   MsgBox "整机库存结转成功！", vbInformation, Forms!usysfrmLogin.Caption

                   DoCmd.Close acForm, "frmZjMonth"

    End If

End Sub
这段代码主要是通过一个追加查询，把表tblZjxx中的jxid，brkc，kcmonth追加到表tblZjkcyj中去，完成库存结转。如果对上面的代码有疑问，可以在VBA的帮助里面搜索一下insert into语句的帮助，通过里面的示例和讲解相信大家都会知道它的用法了。

在关闭按钮中写代码如下：

Private Sub cmdClose_Click()

DoCmd.Close acform,me.name
End Sub
到这里库存结转窗体基本就完成了，我们再回到usysfrmmain窗体的设计视图，在我们添加的按钮的单击事件里写代码如下：

Private Sub cmdJzzj_Click()

DoCmd.OpenForm "frmZjMonth"

End Sub
